Cambridge University Press
978-0-521-83626-5 - Complex Analysis with Mathematica ®
William T. Shaw
Excerpt
More information

1 Why you need complex numbers

Introduction

The complex number system is now such an accepted part of mathematical analysis that it requires some adjustment of your point of view just to ask *why* you need so-called imaginary or complex numbers. But you should understand that there was, at first, considerable resistance to their introduction, even amongst those who felt compelled to invent them! Probably the first person to discuss them was Girolamo Cardano, in his text *Ars magna* (*The Great Art*), published in 1545 (Cardano, 1993). Cardano also was one of the first Western algebraists to cope with the concept of negative numbers, and to introduce negative roots. The additional headache involved in dealing with imaginary numbers was such that he largely kept them out of his book, with the exception of a brief discussion of the solution of the quadratic equation :

$$x(10 - x) = 40 \tag{1.1}$$

Cardano did not cope terribly well with the processes involved in managing this equation – as he put it: ‘putting aside the mental tortures involved, multiply $5 + \sqrt{(-15)}$ by $5 - \sqrt{(-15)}$, making $25 - (-15)$, whence the product is 40’. He went on to add: ‘So progresses arithmetic subtlety the end of which, as is said, is as refined as it is useless.’

One view of this book is therefore that it is devoted to a useless topic, but it is likely that Cardano might have shifted his opinions if he could only have experienced the outcome of his mental tortures. If you wish to experience Cardano's reservations personally, his text is available in translation. Also, an excellent exposition of the history of algebra in the sixteenth century is given by Burton (1995). Cardano pushed his luck in other ways, and through his fascination for astrology he managed to get himself imprisoned for heresy. The author hastens to add that it was not so much the introduction of imaginary numbers that got him into trouble – rather more offence was taken at his publishing a horoscope of Jesus Christ. Controversy surrounded Cardano and others interested in the solution of polynomial equations – the intense competition to understand quadratic, cubic and quartic equations generated considerable rivalry!

This book is devoted to explaining why complex numbers and complex analysis are two of the most useful topics in pure and applied mathematics, physics and engineering. You need them.

1.1 First analysis of quadratic equations

If you wish to understand how complex numbers arise from simple polynomial equations with real coefficients, it is sufficient to analyse the following quadratic equation:

$$ax^2 + bx + c = 0 \tag{1.2}$$

You can rewrite this in the form

$$a(x^2 + (2\,xb)/(2\,a)) + c = 0 \tag{1.3}$$

If you then 'complete the square', you obtain

$$a(x + b/(2\,a))^2 - b^2/(4\,a) + c \;=\; 0 \tag{1.4}$$

Hence, you arrive at:

$$(x + b/(2\,a))^2 = (b^2 - 4\,ac)/(2\,a)^2 \tag{1.5}$$

As long as x is real, the left side of this equation is the square of a real number, and is therefore non-negative. The right side is non-negative if and only if

$$b^2 - 4\,ac \geq 0 \tag{1.6}$$

You can, in this case, take the ordinary square root, to obtain

$$x + b/(2\,a) = \pm\left(\sqrt{(b^2 - 4\,ac)}\right)\big/(2\,a) \tag{1.7}$$

$$x = \left(-b \pm \sqrt{(b^2 - 4\,ac)}\right)\big/(2\,a) \tag{1.8}$$

If, on the other hand,

$$b^2 - 4\,ac < 0 \tag{1.9}$$

the square root cannot be taken in the usual way. The introduction of a quantity i (*Mathematica*® uses a double-struck character for the standard mathematical representation) satisfying

$$i^2 = -1 \tag{1.10}$$

resolves the matter, since then you can write

$$x = \left(-b \pm i\sqrt{(4\,a\,c - b^2)}\right)\big/(2\,a) \tag{1.11}$$

Thus you obtain complex roots of an equation with real coefficients. It is another matter to understand what happens when the coefficients themselves are complex. At first you might wonder if you have to extend the number system still further to cope. It is one of the important results of complex analysis that this is unnecessary. You will discover that all polynomial equations of degree n, with coefficients that are complex numbers, have n roots that are complex numbers. In other words: *Complex numbers are enough*.

Now is a good time to get a grip on the use of *Mathematica* to solve simple equations. If you are completely new to *Mathematica* you may first wish to explore the booklet *Getting Started with Mathematica*… (see your *Mathematica* documentation kit). If you do not wish to explore the use of *Mathematica* to solve equations, skip to the next chapter.

1.2 *Mathematica* investigation: quadratic equations

If you are using this text with a computer, start the *Mathematica* system by clicking (e.g. in the MacOS X 'dock') or double-clicking (on most other systems) on the appropriate icon on your computer system. This section will give you a brief introduction on using *Mathematica* to solve simple quadratic equations. In each case you can just type in the commands given in bold-face **Courier** ('typewriter') font, and enter the command using either Shift Return, RET; Enter, ENTER; or Insert (depending on your operating system). If you are viewing this notebook from a CD-ROM or other electronic source, you can of course just browse the existing material entering some or all of the commands.

Since the solution of equations is one of your main goals, and is a running theme of this book, now is a good time to explore the **Solve** function that is built into *Mathematica*. Although the material of this section is basic, it illustrates an important and basic point about how *Mathematica* 'solves' equations by *returning replacement rules*. You should try the following examples:

```
mysolution = Solve[x^2 - 1 == 0, x]
```

$\{\{x \to -1\}, \{x \to 1\}\}$

Note that you use a double equal sign when you wish to denote *equality* in a *Mathematica* expression representing an equation. (The use of a single equal sign is reserved for assignment.) The solution to this quadratic is now contained in the *Mathematica* expression **mysolution**, and we can ask about this expression:

```
mysolution
```

$\{\{x \to -1\}, \{x \to 1\}\}$

What we really want are the values of x *given* the result contained in **mysolution**. In *Mathematica* the English word 'given' is expressed by the combination **/.** (slash dot):

```
myx = x /. mysolution
```

$\{-1, 1\}$

This gets you a list of the solutions expressed in the variable **myx**. This list contains two elements, and we can extract each one by referring to the position in the list. This is achieved by the use of double square brackets, containing the position in the list of the result of interest:

```
myx[[2]]
```

1

Cambridge University Press
978-0-521-83626-5 - Complex Analysis with Mathematica ®
William T. Shaw
Excerpt
More information

■ Some other real equations with purely real roots

In the following example you go directly to the list of solutions:

```
myxTwo = x /. Solve[x^2 - 2 == 0, x]
```

$$\{-\sqrt{2}, \sqrt{2}\}$$

We can now ask for any particular result:

```
myxTwo[[1]]
```

$$-\sqrt{2}$$

```
myxThree = x /. Solve[x^2 + 2 x - 2 == 0, x]
```

$$\{-1-\sqrt{3}, -1+\sqrt{3}\}$$

```
myxFour = x /. Solve[x^2 - x - 1 == 0, x]
```

$$\left\{\frac{1}{2}\left(1-\sqrt{5}\right), \frac{1}{2}\left(1+\sqrt{5}\right)\right\}$$

If you wish to extract the numerical values of the solution, you use the **N[ ]** function. This can be applied by placing the expression to be numericalized in single square brackets – this is how *Mathematica* expects to see all arguments to functions, including **N[ ]**:

```
N[myxFour]
```

{−0.618034, 1.61803}

You can also apply **N[ ]** ‘afterwards’, using the **//N** construction:

```
myxFour // N
```

{−0.618034, 1.61803}

■ Real equations with purely imaginary roots

In the following examples, *Mathematica* extracts the purely imaginary roots from an equation with real coefficients:

```
myxFive = x /. Solve[x^2 + 1 == 0, x]
```

$$\{-i, i\}$$

```
myxSix = x /. Solve[x^2 + 2 == 0, x]
```

$$\{-i\sqrt{2}, i\sqrt{2}\}$$

```
myxSeven = x /. Solve[x^2 + 12 == 0, x]
```

$$\{-2\,i\sqrt{3}, 2\,i\sqrt{3}\}$$

Note that in *Mathematica*'s standard form and traditional form for expressions, the square root of -1 is denoted by i. You can always ask *Mathematica* for the traditional mathematical form of an expression by using explicit conversion to **TraditionalForm**. In most of this book the traditional form is used for output – it is more elegant than **StandardForm**.

```
StandardForm[myxSix]
```

$$\{-i\sqrt{2}, i\sqrt{2}\}$$

```
TraditionalForm[%]
```

$$\{-i\sqrt{2}, i\sqrt{2}\}$$

Note that **StandardForm** uses an upright font and **TraditionalForm** an italic font. If you are new to *Mathematica*, note that the "per cent" symbol % is a useful shortcut to the last output. In this case there is just some basic tidying up of the spacing. In the second example there is more tidying to present the output essentially as you would write it on paper. If you look under the *Mathematica* Cell Menu you will see that there are menu commands and keyboard shortcuts for converting between:

InputForm – *Mathematica*'s standard pure text representation of expressions;
StandardForm – A compromise between traditional mathematical notation and an unambiguous computer representation;
TraditionalForm – Traditional mathematical notation.

■ Real equations with complex roots

Try the following examples:

```
myxEight = x /. Solve[x^2 + x / 2 + 1 == 0, x]
```

$$\left\{\frac{1}{4}\left(-1 - i\sqrt{15}\right), \frac{1}{4}\left(-1 + i\sqrt{15}\right)\right\}$$

```
N[myxEight]
```

$$\{-0.25 - 0.968246\,i, -0.25 + 0.968246\,i\}$$

It is always a good idea to check that your answers regenerate the original quadratic:

```
Expand[(x - (myxEight[[1]])) (x - (myxEight[[2]]))]
```

$$x^2 + \frac{x}{2} + 1$$

■ Complex equations with complex roots

Although you may not yet know how to treat equations that have complex coefficients, it is perfectly possible (you will read how to deal with this later, in Chapter 2); *Mathematica* does not need any further instruction. Here is a good place to note that the 'Input Form' of the square root of -1 is the capital **I**:

```
myxNine =
 x /. Solve[(1 + I) x^2 + (2 + I) x + 3 - 2 I == 0, x]
```

$$\left\{\left(\frac{1}{4} + \frac{i}{4}\right)\left((-1 + 2\, i) + \sqrt{17}\right), \left(-\frac{1}{4} - \frac{i}{4}\right)\left((1 - 2\, i) + \sqrt{17}\right)\right\}$$

Finally, you can consider the case we started with:

```
myxTen = x /. Solve[a x^2 + b x + c == 0, x]
```

$$\left\{\frac{-b - \sqrt{b^2 - 4\, a\, c}}{2\, a}, \frac{\sqrt{b^2 - 4\, a\, c} - b}{2\, a}\right\}$$

■ Using **Factor** (more advanced)

You have seen how to use **Solve** to solve the equation. You can also use **Factor** to work on the quadratic (or indeed any polynomial) itself, if you are interested in looking at factorizations over only the integers:

```
Factor[2 x^2 - 2]
```

$$2\,(x - 1)\,(x + 1)$$

You can get a list of the factors together with their powers as follows:

```
FactorList[2 x^2 - 2]
```

$$\begin{pmatrix} 2 & 1 \\ x - 1 & 1 \\ x + 1 & 1 \end{pmatrix}$$

However, look what happens when you try the following:

```
Factor[x^2 + 1]
```

$$x^2 + 1$$

You can examine the effect of resetting one of the options:

```
Options[Factor]
```

{Extension → None, GaussianIntegers → False, Modulus → 0, Trig → False}

If we allow Gaussian integers, that is, complex numbers whose real and imaginary parts are both integers, then a different result is obtained:

```
Factor[x^2 + 1, GaussianIntegers -> True]
```

$(x - i)(x + i)$

You can obtain further information about **Factor**, or indeed any *Mathematica* function, by prefixing the function name with a question mark:

```
? Factor
```

```
Factor[poly] factors a polynomial over the integers. Factor[poly, Modulus->
   p] factors a polynomial modulo a prime p. Factor[poly, Extension->
   {a1, a2, ... }] factors a polynomial allowing coefficients
   that are rational combinations of the algebraic numbers ai. …
```

So, for example, if you try the following, you get nowhere:

```
Factor[x^2 + 2, GaussianIntegers -> True]
```

$x^2 + 2$

But if you allow an extension, the factorization proceeds:

```
Factor[x^2 + 2, GaussianIntegers -> True,
    Extension -> {Sqrt[2]}]
```

$\left(\sqrt{2} - i\,x\right)\left(i\,x + \sqrt{2}\right)$

You can build up lists of suitable extensions very easily. Here you make a list of the square roots of the first three primes:

```
mylist = Table[Sqrt[Prime[k]], {k, 3}]
```

$\left\{\sqrt{2}, \sqrt{3}, \sqrt{5}\right\}$

This allows you to play with higher-order polynomials:

```
Factor[(x^2 - 3) (x^2 - 5) (x^2 - 6), Extension -> mylist]
```

$-\left(\sqrt{3} - x\right)\left(\sqrt{5} - x\right)\left(\sqrt{6} - x\right)\left(x + \sqrt{3}\right)\left(x + \sqrt{5}\right)\left(x + \sqrt{6}\right)$

But now we have jumped to higher-order polynomials. It is time to develop some more theory. In the next chapter you will see how to define complex numbers properly, and you will soon be able to explore the solution of cubic, quartic and other polynomial equations.

Exercises

1.1 Using pen and paper only (i.e. not using *Mathematica)* find the solutions of the following quadratic equations:

$$x^2 - 3 = 0$$

$$x^2 + 2x + 1 = 0$$

$$x^2 + 3 = 0$$

$$x^2 + x + 4 = 0$$

1.2 If $a > 0$, what is the minimum value of the expression

$$ax^2 + bx + c$$

and for what value of x does it occur? When is this minumum value negative, and what does this imply about the roots of the quadratic equation

$$ax^2 + bx + c = 0\,?$$

What happens when this minumum value is zero? What happens when it is positive? Interpret your results graphically.

1.3 Suppose that x_1 and x_2 are the roots of the quadratic equation

$$ax^2 + bx + c = 0$$

By writing this in the form

$$a(x - x_1)(x - x_2) = 0$$

show that

$$x_1 x_2 = c/a$$
$$x_1 + x_2 = -b/a$$

Hence write down a quadratic equation with roots $3 + 2i$ and $1 - i$.

1.4 Let f be the function given by:

$$f(x) = \lambda x(1 - x)$$

Without using *Mathematica*, find both solutions of the quadratic equation:

$$x = f(x)$$

Hence, without using *Mathematica*, find all the solutions of the quartic equation:

Cambridge University Press
978-0-521-83626-5 - Complex Analysis with Mathematica ®
William T. Shaw
Excerpt
More information

$$x = f(f(x))$$

Hint: the solutions of the quadratic equation are necessarily solutions of the quartic equation.

1.5 ❄ Use *Mathematica*'s **`Solve`** function to solve the quadratic equations in Exercise 1.1, and compare your solutions.

1.6 ❄ Plot the functions x^2, $x^2 - 1$, and $x^2 + 1$, using the *Mathematica* **`Plot`** function, For example, the first of these may be plotted with:

```
Plot[x^2, {x, -2, 2}]
```

1.7 ❄ Repeat the exercise of problem 1.6, but parametrize the constant by a value **`c`**. Explore the use of an animation to see the results, by trying the following:

```
Do[Plot[x^2 + c, {x, -2, 2}, PlotRange -> {-1, 3}],
 {c, -1, 1, 0.2}]
```

What is the relationship between the quantity $b^2 - 4ac$ in this case, and the location of the curve?

1.8 ❄ Use **`Factor`** to find the real linear and quadratic factors of the expression $x^4 - 1$, and hence find all solutions of the quartic equation $x^4 - 1 = 0$.

1.9 ❄ Using the methods of Exercise 1.8, find all six solutions to the equation $x^6 - 1 = 0$.

CAMBRIDGE

Cambridge University Press
978-0-521-83626-5 - Complex Analysis with Mathematica ®
William T. Shaw
Excerpt
More information

2 Complex algebra and geometry

Introduction

In the first chapter you saw why you need imaginary and complex numbers, by considering the solution of simple quadratic equations. In this chapter you will see how we set up complex numbers in general, and establish their basic algebraic and geometrical properties.

We shall assume that you have some understanding of what is meant by a real number. The exact nature and depth of this understanding will not materially affect the discussion thoughout most of this book, and this is not a book about the fundamentals of real analysis. We should, however, take a moment to remind ourselves what a 'real' number is, before we start defining 'imaginary' and 'complex' numbers. Students of pure mathematics should remind themselves of the details of these matters – there is really nothing for it but to go for a proper mathematical definition, and experience has shown that one needs to be slightly abstract in order to get it right, in the sense that the resulting definition contains all the numbers 'we need'. For a full exposition, complete with proofs, you should consult a text on real analysis, such as that by Rudin (1976). For our purposes it will mostly be sufficient to regard real numbers as being all the points on a line (which we call the real axis) extending to infinity in both directions. This contains positive and negative integers, rational numbers, such as 1/2 and 17/15, simple square roots such as the square root of 2, and other numbers such as π and e. When we come to consider certain results about limits of infinite sequences and series, it will be necessary to call on more formal results from analysis.

2.1 Informal approach to 'real' numbers

For completeness we begin by briefly exploring the necessity of the 'real' number system. In fact, one reason for having 'real' numbers at all is the inadequacy of integers and rationals (ratios of integers) for solving equations. This parallels our need for introducing complex numbers, but at a more basic level. So we begin by considering even simpler equations than those considered in Chapter 1. For example, consider the solution of the following two equations. Although this can be done with 'pen and paper', we carry this out in *Mathematica*, asking for x given (`/.`) the results of the **`Solve`** function. First note that this function returns a list of replacement rules:

```
Solve[2 x == 16, x]
```

$\{\{x \to 8\}\}$

To get the value of x given this result we ask for:

